

4 - IF ... THEN ... ELSE ... (= LA STRUTTURA DI SELEZIONE)

- Esempio 1: `if a*a+b*b=c*c then write ('rettangolo') else write ('non rettangolo')`

In generale, la struttura (*struttura* = istruzione che governa altre istruzioni)

IF condizione THEN istruzione 1 (ELSE istruzione 2)

ordina al computer di:

- i) controllare se è verificata la *condizione*;
- ii) in caso affermativo, eseguire l'*istruzione 1*,
in caso negativo eseguire l'*istruzione 2*
(questa parte con l'**ELSE** può anche mancare)

- Esempio 2:

```
if a*a+b*b=c*c then
    begin
        writeln ('Il triangolo è rettangolo, perché la somma');
        writeln ('dei quadrati di due lati uguaglia il quadrato del terzo');
    end
else write ('Il triangolo non è rettangolo')
```

Notare in questo es. l'uso degli indicatori di INIZIO BLOCCO (**begin**) e FINE BLOCCO (**end**).

- Esempio 3:

if media<4 then media:=4

Questa istruzione potrebbe far parte di un programma nel quale un insegnante indulgente decida di "alzare" al 4, per non infierire, la media di un suo alunno nel caso questa risulti < 4; in caso contrario, cioè se la media è maggiore o uguale a 4, essa non verrà modificata.

L'esempio ribadisce che l'**ELSE non è sempre obbligatorio nell'ambito di una IF**.

ATTENZIONE: prima dell' ELSE non ci vuole mai il "punto-e-virgola" (errore frequente!)
Il compilatore si arresterebbe fornendo il "Syntax error" che segue: *“;” expected but “ELSE” found.*

```
program equazioni_strane_1; uses crt;
var a, b, x: real;
begin
  clrscr;
  writeln ('Risolviamo l''equazione ax=b');
  write ('a = '); readln (a); write ('b = '); readln (b);
  IF a<>0 THEN
    begin
      x:=b/a;
      write ('x = ', x);
    end
  ELSE
    IF b<>0 THEN write ('Eq. Impossibile')
    ELSE write ('Eq. Indeterminata');
  readln;
end.
```

Questo programma ci dà un esempio di due IF ... THEN ... ELSE ... contenute una dentro l'altra, cioè "ANNIDATE"

```
program equazioni_strane_2; uses crt;
var a, b, x: real;
begin
  clrscr;
  writeln ('Risolviamo l''equazione ax=b'); (*NOTA*)
  write ('a = '); readln (a); write ('b = '); readln (b);
  if a<>0 then write ('x = ', b/a);
  if (a=0) AND (b<>0) then write ('Eq. Impossibile');
  if (a=0) AND (b=0) then write ('Eq. Indeterminata');
  readln;
end.
```

L'uso dell'operatore logico AND ci ha permesso di scrivere il programma in modo più semplice.

**Gli operatori logici in Pascal sono:
AND, OR, NOT, XOR
(XOR = disgiunzione esclusiva:
 $p \text{ XOR } q$ è vera se e solo se
è vera una e una sola delle due condizioni p, q)**

Esercizio 1)

Nel 1969 l'uomo è sbarcato per la prima volta sulla Luna.
Scrivi un programma PASCAL che domandi all'utente il suo anno di nascita e fornisca, a seconda dei casi, l'output:

- i) "Quando l'uomo sbarcò sulla Luna tu avevi ... anni" (NOTA)
- ii) "Quando l'uomo sbarcò sulla Luna tu non eri ancora nato: saresti nato ... anni dopo"
- iii) "Caspita, ma sei nato proprio l'anno in cui l'uomo sbarcò sulla Luna!!!"

NOTA
Gli apostrofi nelle stringhe vengono indicati tramite un doppio apice: vedi pag. 9